
1. [bookmark: _Toc100740788]Corrections required
1. [bookmark: _Toc30685521][bookmark: _Toc100740789]Change 1
	Function name
	fl_NR5GC_ModifyPDUSession_RemoveIMS_Media

	Reason for change
	fl_NR5GC_ModifyPDUSession_RemoveIMS_Media is responsible to release radio resources during IMS voice/video calls. Currently, the sequence is that the radio bearer that is going to be removed, is firstly released from SS and then its release signalled to the UE. Since during IMS calls it is expected that DRB traffic is constantly flowing (video packets / comfort noise packets) it may very well happen that while the bearer release has already been applied on the SS side but not yet signalled to the UE a PDU on that bearer is received by SS. This results in an undefined SS behaviour.
To prevent that we propose to modify the sequence such that firstly the bearer release is signalled to UE and then configured on SS.

	Summary of change
	f_NR_RRCReconfigNewDRBWithSSConfig is broken down within fl_NR5GC_ModifyPDUSession_RemoveIMS_Media so the RRC signalling message and the SS configuration can be separated and its order swapped.

	TTCN module
	NR5GC\NR5GC_IMSProcedures.ttcn

	MCC160 Comment
	



Before Change
	    function fl_NR5GC_ModifyPDUSession_RemoveIMS_Media(NR_CellId_Type p_NR_CellId, ListOfMedia_Type p_MediaType) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_SS_Config;
    var template (value) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), IMS_DNN);
    var IntegerList_Type v_DRBIdList := f_NR_GetDRBList_ForIMSPDUSession();
    var template (value) NR_RLC_BearerToReleaseList_Type v_RLC_BearerToReleaseList;
    var NR_DRBInfo_Type v_DRBInfo;


    var template (value) DRB_ToReleaseList v_DRB_ToReleaseList;
    var template (value) QoS_RuleList v_QoS_Rules;
    var template (value) QoS_FlowList v_QoS_Flows;
    var boolean v_RemoveVoice := false;
    var boolean v_RemoveVideo := false;
    var integer i;
    var integer j := 0;

    
    for (i := 0; i < lengthof(p_MediaType); i := i+1) { // @sic R5-217798 sic@
      if (p_MediaType[i] == IPCAN_SpeechCall) {
        v_RemoveVoice := true;
      }
      if (p_MediaType[i] == IPCAN_VideoCall) {
        v_RemoveVideo := true;
      }
    }
    
    // @sic R5-216150 sic@
    if (v_RemoveVoice) { // @sic R5-217798 sic@
      v_QoS_Rules[j] := cs_QoS_DeleteRule('03'O); // @sic R5s210073, R5-213439 sic@
      v_QoS_Flows[j] := cs_QoS_DeleteFlow('000111'B); // @sic R5-213439 sic@
      j := j+1;
    }
    if (v_RemoveVideo) {
      if (v_RemoveVoice) {
        v_QoS_Rules[j] := cs_QoS_DeleteRule('04'O);
      } else {  // @sic R5-217798 sic@ FFS: error in prose
        v_QoS_Rules[j] := cs_QoS_ModifyRule('04'O);
      }
        v_QoS_Flows[j] := cs_QoS_DeleteFlow('001000'B);
    }

    // build components of Reconfiguration message
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId,
                                                                         omit,
                                                                         -, -, -, -,
                                                                         cs_QoS_Rules(v_QoS_Rules, '7A'O),
                                                                         cs_QoSFlowDescr(v_QoS_Flows));

    j := 0;  // reset index
    // @sic R5-211171 sic@
    for (i := 0; i < lengthof(v_DRBIdList); i := i+1) {
      v_DRBInfo := valueof(f_NR_GetDRBInfo_FromId(v_DRBIdList[i]));
      if (not (v_DRBInfo.DefaultDRB)) {
        if ((v_RemoveVoice and (v_DRBInfo.SDAP_Config.mappedQoS_FlowsToAdd[0] == 7)) or
             v_RemoveVideo and (v_DRBInfo.SDAP_Config.mappedQoS_FlowsToAdd[0] == 8)) { // @sic R5-217798 sic@
          v_DRB_ToReleaseList[j] := v_DRBInfo.DRB_Id;
          v_RLC_BearerToReleaseList[j] := v_DRBInfo.RLC_Config.logicalChannelIdentity;
          v_SS_Config[j] := cs_NR_SS_RadioBearer_Release(cs_NR_RadioBearerId_DRB(v_DRBInfo.DRB_Id));

          j := j+1;
        }
      }
    }

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit, omit, v_DRB_ToReleaseList);
    v_CellGroupConfig := cs_38508_CellGroupConfig(f_NR_CellInfo_Get_CellGroupId(p_NR_CellId), omit, v_RLC_BearerToReleaseList);

    // The SS transmits an RRCReconfiguration message with a drb-ToReleaseList and PDU SESSION MODIFICATION COMMAND



    f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                       v_RadioBearerConfig,
                                       v_SS_Config,
                                       v_CellGroupConfig,
                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)},
                                       -, -, false);  // @sic R5-216150 sic@

    
    interleave {  // @sic R5-216150 sic@
        []SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId,
                                            tsc_NR_RbId_SRB1,
                                            cr_38508_RRCReconfigurationComplete)) {}
        []SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId,
                                            tsc_NR_RbId_SRB2,
                                            cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp{}
    }




    if (not f_Check_NG_PDUSessionModificationComplete (v_ReceivedAsp.Signalling.Nas[0].Pdu, v_GSM_MobilityInfo.SessionId)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Modification Complete failed");
    }
    if (v_RemoveVoice) { // @sic R5-217798 sic@
      f_IP_RtpRtcpLoopback(IP, valueof(v_GSM_MobilityInfo.PdnIndex), tsc_IMS_MediaPort_M1, omit);  // @sic R5s210363 sic@
    }
    if (v_RemoveVideo) { // @sic R5-217798 sic@
      f_IP_RtpRtcpLoopback(IP, valueof(v_GSM_MobilityInfo.PdnIndex), tsc_IMS_MediaPort_M2, omit);  // @sic R5s210363 sic@
    }
  }




After Change
	  
  function fl_NR5GC_ModifyPDUSession_RemoveIMS_Media(NR_CellId_Type p_NR_CellId, ListOfMedia_Type p_MediaType) runs on NR5GC_PTC
  {
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
    var template (value) NR_RadioBearerList_Type v_SS_Config;
    var template (value) GSM_MobilityInfo_Type v_GSM_MobilityInfo := f_Get_PDUSessionForDNNType(f_NR5GC_MobileInfo_GetSessionInfoList(), IMS_DNN);
    var IntegerList_Type v_DRBIdList := f_NR_GetDRBList_ForIMSPDUSession();
    var template (value) NR_RLC_BearerToReleaseList_Type v_RLC_BearerToReleaseList;
    var NR_DRBInfo_Type v_DRBInfo;
    var NR_DRBInfoList_Type v_ExistingDRBs;
    var IntegerList_Type v_RemovedDRBList := {};
    var template (value) DRB_ToReleaseList v_DRB_ToReleaseList;
    var template (value) QoS_RuleList v_QoS_Rules;
    var template (value) QoS_FlowList v_QoS_Flows;
    var boolean v_RemoveVoice := false;
    var boolean v_RemoveVideo := false;
    var integer i, v_release_index;
    var integer j := 0;
    var template (omit) RRCReconfiguration_v1530_IEs v_V1530Ext;
    
    for (i := 0; i < lengthof(p_MediaType); i := i+1) { // @sic R5-217798 sic@
      if (p_MediaType[i] == IPCAN_SpeechCall) {
        v_RemoveVoice := true;
      }
      if (p_MediaType[i] == IPCAN_VideoCall) {
        v_RemoveVideo := true;
      }
    }
    
    // @sic R5-216150 sic@
    if (v_RemoveVoice) { // @sic R5-217798 sic@
      v_QoS_Rules[j] := cs_QoS_DeleteRule('03'O); // @sic R5s210073, R5-213439 sic@
      v_QoS_Flows[j] := cs_QoS_DeleteFlow('000111'B); // @sic R5-213439 sic@
      j := j+1;
    }
    if (v_RemoveVideo) {
      if (v_RemoveVoice) {
        v_QoS_Rules[j] := cs_QoS_DeleteRule('04'O);
      } else {  // @sic R5-217798 sic@ FFS: error in prose
        v_QoS_Rules[j] := cs_QoS_ModifyRule('04'O); 
      }
        v_QoS_Flows[j] := cs_QoS_DeleteFlow('001000'B);
    }

    // build components of Reconfiguration message
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand(v_GSM_MobilityInfo.SessionId,
                                                                         omit,
                                                                         -, -, -, -,
                                                                         cs_QoS_Rules(v_QoS_Rules, '7A'O),
                                                                         cs_QoSFlowDescr(v_QoS_Flows));

    j := 0;  // reset index
    // @sic R5-211171 sic@
    for (i := 0; i < lengthof(v_DRBIdList); i := i+1) {
      v_DRBInfo := valueof(f_NR_GetDRBInfo_FromId(v_DRBIdList[i]));
      if (not (v_DRBInfo.DefaultDRB)) {
        if ((v_RemoveVoice and (v_DRBInfo.SDAP_Config.mappedQoS_FlowsToAdd[0] == 7)) or
             v_RemoveVideo and (v_DRBInfo.SDAP_Config.mappedQoS_FlowsToAdd[0] == 8)) { // @sic R5-217798 sic@
          v_DRB_ToReleaseList[j] := v_DRBInfo.DRB_Id;
          v_RLC_BearerToReleaseList[j] := v_DRBInfo.RLC_Config.logicalChannelIdentity;
          v_SS_Config[j] := cs_NR_SS_RadioBearer_Release(cs_NR_RadioBearerId_DRB(v_DRBInfo.DRB_Id));
          v_release_index := j;
          j := j+1;
        }
      }
    }

    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit, omit, v_DRB_ToReleaseList);
    v_CellGroupConfig := cs_38508_CellGroupConfig(f_NR_CellInfo_Get_CellGroupId(p_NR_CellId), omit, v_RLC_BearerToReleaseList);

    // The SS transmits an RRCReconfiguration message with a drb-ToReleaseList and PDU SESSION MODIFICATION COMMAND

/*REMOVED
    f_NR_RRCReconfigNewDRBWithSSConfig(p_NR_CellId,
                                       v_RadioBearerConfig,
                                       v_SS_Config,
                                       v_CellGroupConfig,
                                       { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)},
                                       -, -, false);  // @sic R5-216150 sic@
*/    

    //Remove DRB that is being released and store the new info
    v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
    v_RemovedDRBList := valueof (v_RadioBearerConfig.drb_ToReleaseList);
    v_ExistingDRBs := f_NR_DRBInfo_RemoveEntry (v_ExistingDRBs, v_RemovedDRBList[v_release_index]);
    f_NR5GC_MobileInfo_SetDRBInfoList(v_ExistingDRBs);

    //Prepare RRC and send it
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));
    f_NR_SendRRCReconfiguration (p_NR_CellId, tsc_NR_RbId_SRB1, v_RadioBearerConfig, v_V1530Ext, { cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)}, -, -, false);

    interleave {  // @sic R5-216150 sic@
        []SRB.receive(car_NR_SRB_RrcPdu_IND(p_NR_CellId,
                                            tsc_NR_RbId_SRB1,
                                            cr_38508_RRCReconfigurationComplete)) {}
        []SRB.receive(car_NR_SRB_NasPdu_IND(p_NR_CellId,
                                            tsc_NR_RbId_SRB2,
                                            cr_NG_NAS_IndWithPiggybacking((tsc_SHT_IntegrityProtected_Ciphered,tsc_SHT_IntegrityProtected)))) -> value v_ReceivedAsp{}
    }

    //And now when it is released on the UE, it is safe to do the same on SS side
    f_NR_SS_CommonRadioBearerConfig(p_NR_CellId, v_SS_Config);

    if (not f_Check_NG_PDUSessionModificationComplete (v_ReceivedAsp.Signalling.Nas[0].Pdu, v_GSM_MobilityInfo.SessionId)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "PDU Session Modification Complete failed");
    }
    if (v_RemoveVoice) { // @sic R5-217798 sic@
      f_IP_RtpRtcpLoopback(IP, valueof(v_GSM_MobilityInfo.PdnIndex), tsc_IMS_MediaPort_M1, omit);  // @sic R5s210363 sic@
    }
    if (v_RemoveVideo) { // @sic R5-217798 sic@
      f_IP_RtpRtcpLoopback(IP, valueof(v_GSM_MobilityInfo.PdnIndex), tsc_IMS_MediaPort_M2, omit);  // @sic R5s210363 sic@
    }
  }



